Het aanmaken van een mailing

door Walther Musch


Inleiding

Het aanmaken van een Word document in Visual Basic dat gebruikt kan worden om een mailing te verzorgen.

Het artikel ‘Automation met VB en MS Office’ in het vorige kwartaalnummer deed mij terugdenken aan een functie die ik begin vorig jaar heb gemaakt. Bij een standaard adressenbestand in Visual Basic met een MS Access database wilden de gebruikers graag de mogelijkheid om direct vanuit het programma een mailing weg te sturen. 

Na het bekijken van de functie in Word (MailMerge of Samenvoegen)  bleek dat Word in feite twee documenten gebruikt: het data-document waarin een simpele tabel de gegevens staan. En het bron-document waarin de standaard tekst en velden zijn geplaatst. In eerste instantie moest de functie het data-document kunnen aanmaken. Het bron-document werd eenmalig in Word zelf aangemaakt. Het uiteindelijke samenvoegen en printen wordt in de applicatie uitgevoerd. 

Stap 1.
Toevoegen van de References en aanmaken van het project

Stap 2.
Aanmaken van data-document

Stap 3.
Uitvoeren van het samenvoegen

Stap 1:

Start een nieuw project (name = prjMailMerge). Voeg hieraan het object library van Word toe aan het project (Tools – References): vink de betreffende objectlibrary af: Microsoft Word 8.0 Object Library. Omdat we ook gebruik gaan maken van een MS Access database moet ook het object library Microsoft DAO 2.5/3.51 Compatibility Library worden toegevoegd.

Plaats op het standaard aangemaakte form (name = frmMailMerge) twee commandoknoppen (1. Name = cmdMakeData; caption = &Aanmaken 2. Name = cmdMailMerge; caption = &Printen).  Voeg een module toe (name = modMailMerge). De module wordt gebruikt omdat er eigenlijk zo weinig mogelijk code op een form moet worden geplaatst. In plaats daarvan worden de procedures en functies in de module gezet. Op deze manier kan er vanuit meerdere forms gebruik worden gemaakt van dezelfde code.

In dit voorbeeld wordt verder gebruik gemaakt van de nwind.mdb welke te vinden is in de standaard directory waar VB.exe is geïnstalleerd. Alle connecties worden in code uitgevoerd (er wordt geen gebruik gemaakt van de data-control). De SQL om een aantal gegevens op te halen is in het voorbeeld ‘hard-coded’. Bij een echte applicatie moet de gebruiker op de een of andere manier de mogelijkheid hebben om een selectie aan te geven. Ook de namen en locaties van de bestanden staan in code. Daarvoor zou een simpele dialoog control voor kunnen worden gebruikt.

In de frmMailMerge_Load event wordt een verwijzing naar modMailMerge.SetDefault gemaakt. In deze procedure worden een aantal standaard variabelen gezet zoals het openen van de database en de recordset. Ook worden hier – voorlopig voor het voorbeeld – de bestandsnamen gezet. 

'----------------------------------------------------

'

'

Option Explicit

'

Private Declare Sub Sleep Lib _

"kernel32" (ByVal dwMilliseconds As Long)

Private clWord As Word.Application

Private db As Database

Private rs As Recordset

Private strNAW As String

Private strSource As String

Private intRows As Integer

Public Sub SetDefault()

    If Not ConnectDatabase("c:\program files\devstudio\vb\nwind.mdb") Then

        Call MsgBox("error",vbOkOnly,App.ExeName)

        Unload frmMailMerge

    Else

        strNAW = Environ("TEMP") + "\" + _

Format(Now, "ddmmyyhhmm") + ".doc"

        strSource = Environ("TEMP") + "\Example.doc"

    End If

End Sub

Private Function ConnectDatabase(ByRef strName _

As String) As Boolean

    Dim strSQL As String

    '

    On Error GoTo ConnectDatabase_FAIL

    strSQL = "SELECT Suppliers.CompanyName, " + _

" Suppliers.Address, Suppliers.City, “ + _

" Suppliers.Region, Suppliers.PostalCode," + _

" Suppliers.Country FROM Suppliers " + _


“ WHERE Suppliers.Country LIKE 'S*'"

    Set db = OpenDatabase(strName)

    Set rs = db.OpenRecordset(strSQL, dbOpenSnapshot)

    ConnectDatabase = True

    Exit Function

ConnectDatabase_FAIL:

    ConnectDatabase = False

End Function

In de frmMailMerge_Unload event maken we een verwijzing naar modMailMerge.CloseApplication Hier sluiten we nog de open database connectie en een eventuele bestaande Word-object.

Public Sub CloseApplication()

    rs.Close

    db.Close

    Set clWord = Nothing

End Sub

Stap 2.

Onder de click_event van cmdMakeData wordt een verwijzing naar modMailMerge.MakeData gemaakt. In deze procedure openen we eerst een object van Word. (Function StartWord). 

Public Function StartWord() As Boolean

    Dim iTries As Integer

    iTries = 0

    On Error GoTo ErrStartWord

    Set clWord = New Word.Application

    StartWord = True

    Exit Function

ErrStartWord:

    Screen.MousePointer = vbNormal

    Select Case Err.Number

    Case 440

        iTries = iTries + 1

        If iTries < 5 Then

            Set clWord = New Word.Application

            Call Sleep(5)

            Resume

        Else

            Err.Raise Number:= vbObjectError + _

28765, Description:= "Cannot " + _

" start Word (again?)"

            Set clWord = Nothing

            StartWord = False

        End If

    Case Else

        Call MsgBox(CStr(Err) + " " + Error, _ 

vbOKOnly, App.EXEName)

        clWord.Quit

        Set clWord = Nothing

        StartWord = False

    End Select

End Function

Als dat is gelukt dan wordt het aantal benodigde rijen opgehaald. Dit is gelijk aan het aantal records in the recordset + 1. Deze ene extra rij is om de veldnamen van het data-document te plaatsen. 

Eerst wordt een nieuw document aangemaakt. Deze wordt bewaard en vervolgens opnieuw geopend. (Op de een of andere manier moet het op deze manier worden uitgevoerd. Wordt het document niet eerst bewaard dan kunnen er geen bewerkingen op worden uitgevoerd?) Daarna wordt er een tabel structuur aangemaakt. De eerste rij bevat de veldnamen van het data-document. De daaropvolgende rijen bevatten de gegevens. Tenslotte wordt de aangemaakte en gevulde tabel in het data-document definitief bewaard.

Public Sub MakeData()

'making of the datasource Word document

    Dim xWord As Word.Document

    Dim bX As Range

    Dim intC As Integer

    Dim intX As Integer

    Dim intCols As Integer

    On Error GoTo MakeData_FAIL

 'get the number of fields and the number 

 'of records

    intCols = 6

    rs.MoveLast

    intRows = rs.RecordCount

    Screen.MousePointer = vbHourglass

    If Not modMailMerge.StartWord Then _

GoTo ResumeMakeData_FAIL

    'make new document and save this

    Set xWord = clWord.Documents.Add

    xWord.SaveAs strNAW

    'open newly made document

    Call clWord.Documents.Open(strNAW)

    'make a table with # rows and # coloms

    Set bX = clWord.ActiveDocument.Range _

(Start:=0, End:=0)

    clWord.ActiveDocument.Tables.Add bX, _

intRows + 1, intCols

    'set the text in the proper cells

    clWord.ActiveDocument.Tables.Item(1). _

Cell(1, 1).Range.InsertAfter "Naam"

    clWord.ActiveDocument.Tables.Item(1). _

Cell(1, 2).Range.InsertAfter "Adres"

    clWord.ActiveDocument.Tables.Item(1). _

Cell(1, 3).Range.InsertAfter "Plaats"

    clWord.ActiveDocument.Tables.Item(1). _

Cell(1, 4).Range.InsertAfter "Regio"

    clWord.ActiveDocument.Tables.Item(1). _

Cell(1, 5).Range.InsertAfter "Postcode"

    clWord.ActiveDocument.Tables.Item(1). _

Cell(1, 6).Range.InsertAfter "Land"

    '

    rs.MoveFirst

    For intC = 1 To intRows

        For intX = 0 To 5

clWord.ActiveDocument.Tables.Item(1). _

Cell(intC + 1, intX + 1). _ 

Range.InsertAfter "" & _

rs.Fields(intX).Value

        Next intX

        rs.MoveNext

    Next intC

    'save the document under the same name

    xWord.Save

ResumeMakeData_FAIL:

    Set xWord = Nothing

    'you have one document open so you have to 

    'close it before you can exit Word

    Call clWord.ActiveDocument. _

Close(wdDoNotSaveChanges)

    Call clWord.Quit

    Set clWord = Nothing

    Screen.MousePointer = vbNormal

    Exit Sub

MakeData_FAIL:

    If Err.Number = &H80010001 Then

        Call MsgBox("Action canceled.", _

vbOKOnly, App.EXEName)

    Else

        Call MsgBox(CStr(Err) & " " & _ 

Error, vbOKOnly, "MakeData")

    End If

    Resume ResumeMakeData_FAIL

End Sub

Stap 3.

Het click_event van cmdMailMerge verwijst naar de procedure modMailMerge.PrintMailMerge Na het aanmaken van een nieuw Word-object wordt het eerder gemaakte bron-document geopend. Vervolgens wordt het gevulde data-document aangeroepen. Met het juiste gebruik van de Word VBA commando’s worden beide documenten samengevoegd en tenslotte uitgeprint. Word wordt vervolgens weer netjes (transparant) afgesloten.

Public Sub PrintMailMerge()

'mailmerge & printing in Word

    On Error GoTo PrintMailMerge_FAIL

    Screen.MousePointer = vbHourglass

    If Not modMailMerge.StartWord Then _

GoTo PrintMailMerge_EXIT

    'Open document Example.doc

    Call clWord.Documents.Open(strSource)

    'Set the datasource

    Call clWord.ActiveDocument. _ 

MailMerge.OpenDataSource(strNAW)

    If clWord.ActiveDocument.MailMerge. _

DataSource.Name = "" Then

        Call MsgBox("no datasource!", _

vbOKOnly, App.EXEName)

        GoTo PrintMailMerge_FAIL

    End If

    If clWord.ActiveDocument.MailMerge. _

State = wdMainAndDataSource Then _

        clWord.ActiveDocument.MailMerge.Execute

    If clWord.ActiveDocument.MailMerge. _

State = wdMainAndSourceAndHeader Or _

       
clWord.ActiveDocument.MailMerge. _

State = wdMainAndDataSource Then

With clWord.ActiveDocument.MailMerge. _

DataSource

            .FirstRecord = 1

            .LastRecord = intRows

        End With

    End If

    'printing but wait until printing is finished

 
'before going on

    Call clWord.ActiveDocument.PrintOut(False)

PrintMailMerge_EXIT:

    On Error Resume Next

    'because you have TWO documents open in Word: 

    'you must close both documents before closing Word

    Call clWord.ActiveDocument. _

Close(wdDoNotSaveChanges)

    Call clWord.ActiveDocument. _

Close(wdDoNotSaveChanges)

    Call clWord.Quit

    Set clWord = Nothing

    Screen.MousePointer = vbNormal

    Exit Sub

PrintMailMerge_FAIL:

    If Err.Number = &H80010001 Then

        Call MsgBox("Action Canceled.", _

vbOKOnly, App.EXEName)

    Else

        Call MsgBox(CStr(Err) + " " + Error, _

vbOKOnly, App.EXEName)

    End If

    Resume PrintMailMerge_EXIT

End Sub

Zo is met wat regels code een directe transparante verbinding gemaakt tussen de VisualBasic applicatie en Word. Mogelijkheden tot verbeteringen en uitbreidingen zijn voldoende aanwezig. Het zou mooi zijn als gebruikers in de applicatie aan kunnen geven aan welke criteria de uit te voeren data moet voldoen. (Op de site vindt u daar wel wat aanzetten toe. In het volgende kwartaalnummer krijgt u hier een werkbaar voorbeeld van.) Ook het benoemen van de benodigde Word-documenten zou netjes via een dialoog control kunnen lopen. (Ook hiervoor zijn voorbeelden te vinden op de site.) En tenslotte zou het mooi zijn als u binnen uw Visual Basic applicatie het bron-document kunt aanmaken en bewerken (maar dat is eigenlijk oude koek…)

Deze procedure is geschreven in het voorjaar van 1998. Het moest - zoals altijd - snel en makkelijk. Ik heb dan ook niet verder gekeken naar misschien wel betere manieren om hetzelfde doel te bereiken. Mocht u dan ook nuttige wijzigingen en/of verbeteringen hebben dan hoor ik dat graag.

Walther Musch

http://www.kather.net/VisualBasicSource
Het volledige uitgewerkte voorbeeld in code kunt u downloaden als Publish/MailMerge_v1.zip

